

APPLE II SERIES

StripWare™

CLASSIC GAMES

Tim Hartnell

Contains 5 programs:

- Checkers
- Awari
- Gomoku
- Las Vegas High
- Reversi



Package is not returnable if plastic wrap is cut or removed.

TIM HARTNELL, author of such current bestselling computer books as *Creating Adventure Games*, is an Australian journalist who enjoys writing on any computer-related topic. He has written more than 30 books on personal computers, including many game books.

Reprinted with permission of
Tim Hartnell and Ballantine Books

Apple and Applesoft are registered trademarks of
Apple Computer, Inc.

StripWare and Softstrip are trademarks of
Cauzin Systems, Inc.

APPLE II SERIES

StripWare™

CLASSIC GAMES

Tim Hartnell

Contains 5 programs:

- Checkers
- Awari
- Gomoku
- Las Vegas High
- Reversi



Package is not returnable if plastic wrap is cut or removed.

Softstrip™
COMPUTER-READABLE PRINT

Suggested Retail Price

\$12.98

- Always someone to challenge in **CHECKERS**
- Discover the ancient strategy of **AWARI**
- Five-in-a-row wins the game of **GOMOKU**
- Try the **LAS VEGAS HIGH** slot machine
- The game constantly shifts in **REVERSI/OTHELLO™**

Cauzin ©1985

CAUZIN SYSTEMS INC.

835 SOUTH MAIN STREET, WATERBURY, CT 06706

Welcome to The Cauzin Family

Welcome to the world of data strips. You now own a piece of a revolutionary technological advancement that bridges the gap between print and electronic media more quickly and easily than ever before.

Each data strip package contains selected highlights from among the best programs each publisher has to offer from articles, books, and software.

We at Cauzin Systems, Inc., hope you'll enjoy this software package, and believe you'll find data strips to be very practical and useful.

Your comments and ideas on this package or any additional applications you'd like to suggest, would be greatly appreciated. Please send your comments or remarks to:

Cauzin Publishing
835 South Main Street
Waterbury, Conn. 06706

© 1985



CLASSIC GAMES

by Tim Hartnell

CHECKERS

The game of checkers has a long and honorable history. R.C. Bell (in his book *DISCOVERING OLD BOARD GAMES*, Shire Publications, Aylesbury, UK, 1980) says it was invented around 1100, “probably in the south of France, using Backgammon tablemen on a checkered chessboard.”

The *ENCYCLOPEDIA OF SPORTS, GAMES AND PASTIMES* (Fleetway House, London, c. 1935) puts it much further back in time: “Forms of it were known in ancient Egypt, Greece, and Rome, while the game was played in the mid-seventeenth century much as it is today.”

From *GAMES OF THE WORLD*, edited by Frederic V. Grunfeld (Ballantine Books, 1975) more information regarding the origins and evolution of checkers can be found.

The pieces took the name of the queen in French medieval chess, the fers, and the game was known as fierges. Later the chess queen was called dame and French checkers became the jeu de dames as it is still known today. Originally the capture of pieces in a checkers game was optional, as in chess, but early in the sixteenth century compulsory capture of pieces was introduced. Any piece which neglected to make a possible capture was ‘huffed,’ or removed from the board. This enlivened the game and increased its appeal.

Regardless of its age, however, checkers has remained a very popular game around the world, with many European countries having regional variations on the basic play. Continental draughts (checkers is generally known as draughts outside the United States), for example, is played on a board of 100 squares with each player starting the game with 20 pieces.


The CHECKERS program you’ll find here is a version you are probably most familiar with. It plays swiftly, and reasonably well, although its lack of end-game strategy often leads to a dramatic collapse in the final moments of a game.

CHECKERS is played between you and the computer. Each of you is attempting to take (that is to jump over to capture, then remove from the board) the other player’s pieces, or to confine the opponent’s pieces so no more moves are possible.

The game is generally played on a board with 64 squares, that are alternately light and dark. In this program, the board is a series of dots, with your pieces shown as H’s (for human) and the machine’s pieces as C’s (for computer).

When the game begins, you’re at the bottom of the screen, playing up, and the computer is at the top playing down. The program offers you the first move, or you can decline and the computer goes first. Enter moves as a two-digit number, vertical coordinate (1-8) first and then the horizontal coordinate (1-8). Thus, “11” is the lower left-hand corner and “18” is the lower right-hand.

All moves must be made along diagonals. Individual pieces can only move forward, that is toward the opponent’s starting side. Pieces are converted into “kings” when they reach the back rank across the board. Your kings are shown as K’s, and the computer’s kings are dollar signs (\$).




Kings can move either forward or back. After capturing a piece by jumping over it into the vacant square immediately behind the captured piece, you can move again if there is a further capture that can be made.



Reading CHECKERS

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN CHECKERS. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



CHECKERS

by Tim Hartnell

Giant Book of Computer Games

Ballantine Books

Copyright 1983, 1984. All rights reserved.

CHECKERS

```

10 REM CHECKERS
11 REM
12 REM FROM TIM HARTNELL'S
13 REM GIANT BOOK OF
14 REM COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 GOSUB 1010
30 PRINT : PRINT : PRINT : PRINT
  "DO YOU WANT THE FIRST MOVE?
  (Y/N) "; GET A$
40 IF A$ = "Y" THEN 70
50 GOTO 90
60 REM *****
70 GOSUB 600
80 GOSUB 760
90 GOSUB 600
100 GOSUB 130
110 GOTO 70
120 REM *****
130 FOR X = 1 TO 10: S(X) = 0: NEXT
  X
140 SC = 0: A = 89
150 A = A - 1
160 IF Q(A) < > C AND Q(A) < >
  CK THEN 240
170 B = 0: IF A < 29 THEN B = 2
180 B = B + 1
190 M = A + N(B)
200 IF M > 88 OR M < 11 THEN 240

210 IF (Q(M) = H OR Q(M) = HK) AND
  Q(M + N(B)) = E THEN 280
220 IF Q(M) = E THEN IF (Q(M -
  11) < > H AND Q(M - 11) < >
  HK) THEN IF (Q(M - 9) < >
  H AND Q(M - 9) < > HK) AND
  Q(M + 9) < > HK THEN IF ((
  Q(M + 22) < > HK OR Q(M + 1
  8) < > HK) AND (Q(M + 9) <
  > C OR Q(M + 9) < > CK OR
  Q(M + 11) = C OR Q(M + 11) =
  CK)) AND Q(M + 11) < > HK THEN
  GOSUB 400
230 IF B < 2 OR (Q(A) = CK AND B
  < 4) THEN 180
240 IF A > 11 THEN 150
250 FL = 0: IF Q(22) = C OR Q(24)
  = C OR Q(26) = C OR Q(28) =
  C THEN GOSUB 1270
260 IF FL = 1 THEN 570
270 GOTO 420
280 Q(M + N(B)) = Q(A): Q(M) = E: Q
  (A) = E
290 CO = CO + 1
300 GOSUB 600
310 A = M + N(B)
320 B = 0
330 B = B + 1
340 IF (A + 2 * N(B) < 11 OR A +
  2 * N(B) > 88) AND B < 4 THEN
  330
350 M = A + N(B)
360 IF Q(M) = C AND B > 3 THEN RETURN
370 IF (Q(M) = H OR Q(M) = HK) AND
380 IF B < 2 OR (Q(A) = CK AND B
  < 4) THEN 330
390 RETURN
400 IF SC < 10 THEN SC = SC + 1
410 S(SC) = 100 * A + B + 20: RETURN

420 IF SC = 0 THEN 470
430 XC = INT ( RND (1) * SC) + 1

440 A = INT (S(XC) / 100)
450 M = A + N(S(XC) - 100 * A - 2
  0)
460 GOTO 570
470 SC = SC + 1: A = INT ( RND (1
  ) * 88) + 1
480 IF Q(A) < > C AND Q(A) < >
  CK THEN 550
490 B = 0
500 B = B + 1
510 M = A + N(B)
520 IF M > 88 OR M < 11 THEN 540

530 IF Q(M) = E THEN 570
540 IF B < 2 OR Q(A) = CK AND B <
  4 THEN 500
550 IF SC < 300 THEN 470
560 PRINT : PRINT "I CONCEDE THE
  GAME.": END
570 Q(M) = Q(A): Q(A) = E
580 RETURN
590 REM *****
600 HOME : PRINT : PRINT
610 HTAB 9: PRINT "          COMPUTE
  R: "CO
620 HTAB 9: PRINT "          HUMA
  N: "HU: PRINT : PRINT
630 HTAB 9: PRINT "    1 2 3 4 5
  6 7 8"
640 FOR F = 80 TO 10 STEP - 10
650 HTAB 9: PRINT F / 10; " ";
660 FOR G = 1 TO 8: PRINT CHR$
  (Q(F + G)); " ";: NEXT G
670 PRINT " ";: PRINT F / 10: NEXT
  F
680 HTAB 9: PRINT "    1 2 3 4 5
  6 7 8"

```

```

690 IF CO = 12 OR HU = 12 THEN 7
    10
700 RETURN
710 IF HU = 12 THEN PRINT "YOU
    HAVE WON."
720 IF CO = 12 THEN PRINT "I HA
    VE WON."
730 PRINT "THANKS FOR THE GAME."
    : END
740 REM *****
750 REM 99 TO CONCEDE
760 PRINT : PRINT
770 PRINT "ENTER YOUR MOVE (99 T
    O CONCEDE)"
780 INPUT "FROM ";A
790 IF A = 99 THEN GOTO 730
800 IF Q(A) < > H AND Q(A) < >
    HK THEN 780
810 INPUT "TO ";B
815 IF B = A - 9 OR B = A - 11 OR
    B = A + 9 OR B = A + 11 THEN
    820
816 IF (B = A - 18 OR B = A - 22
    OR B = A + 18 OR B = A + 22
    ) AND (Q((A + B) / 2) = C OR
    Q((A + B) / 2) = CK) THEN 82
    0
817 GOTO 810
820 IF Q(B) < > E THEN 810
830 Q(B) = Q(A):Q(A) = E
840 REM *****
850 FOR T = 11 TO 17: IF Q(T) =
    C THEN Q(T) = CK
860 NEXT T
870 FOR T = 82 TO 88: IF Q(T) =
    H THEN Q(T) = HK
880 NEXT T
890 REM *****
900 IF ABS (A - B) < 12 THEN
    RETURN
910 TY = RND (1)
920 IF TY < .3 THEN PRINT "GOOD
    MOVE!": GOSUB 1340
930 IF TY > .7 THEN PRINT "GOT
    ME!": GOSUB 1340
940 HU = HU + 1:Q((A + B) / 2) =
    E: GOSUB 600
950 FOR T = 82 TO 88: IF Q(T) =
    H THEN Q(T) = HK
960 NEXT T
970 PRINT : PRINT "CAN YOU JUMP
    AGAIN (Y/N)? ";: GET A$: PRINT
980 IF A$ < > "Y" THEN RETURN

990 A = B: GOTO 800
1000 REM *****
1010 REM INITIALIZE
1020 HOME : VTAB 3: HTAB 16: PRINT
    "CHECKERS"
1040 PRINT : PRINT
1050 HTAB 13: PRINT "PLEASE STAN
    D BY"
1060 DIM Q(110),N(4),S(10)
1070 H = ASC ("H"):HK = ASC ("K
    ")
1080 C = ASC ("C"):CK = ASC ("S
    ")
1090 E = 32:OF = - 99
1100 FOR M = 1 TO 99:Q(M) = OF: NEXT
    M
1110 FOR M = 1 TO 64
1120 READ D: READ G
1130 Q(D) = G: NEXT M
1140 DATA 81,46,82,67,83,46,84,6
    7,85,46,86,67,87,46
1150 DATA 88,67,71,67,72,46,73,6
    7,74,46,75,67,76,46
1160 DATA 77,67,78,46,61,46,62,6
    7,63,46,64,67
1170 DATA 65,46,66,67,67,46,68,6
    7,51,32,52,46
1180 DATA 53,32,54,46,55,32,56,4
    6,57,32,58,46
1190 DATA 41,46,42,32,43,46,44,3
    2,45,46,46,32
1200 DATA 47,46,48,32,31,72,32,4
    6,33,72,34,46,35,72
1210 DATA 36,46,37,72,38,46,21,4
    6,22,72,23,46,24,72
1220 DATA 25,46,26,72,27,46,28,7
    2,11,72,12,46,13,72
1230 DATA 14,46,15,72,16,46,17,7
    2,18,46
1240 FOR M = 1 TO 4: READ X:N(M)
    = X: NEXT M
1250 DATA -11,-9,11,9
1260 CO = 0:HU = 0: RETURN
1270 IF Q(22) = C AND Q(11) = E THEN
    A = 22:M = 11:FL = 1: RETURN
1280 IF Q(22) = C AND Q(13) = E THEN
    A = 22:M = 13:FL = 1: RETURN
1290 IF Q(24) = C AND Q(13) = E THEN
    A = 24:M = 13:FL = 1: RETURN
1300 IF Q(24) = C AND Q(15) = E THEN
    A = 24:M = 15:FL = 1: RETURN
1310 IF Q(26) = C AND Q(15) = E THEN
    A = 26:M = 15:FL = 1: RETURN
1320 IF Q(26) = C AND Q(17) = E THEN
    A = 26:M = 17:FL = 1: RETURN
1330 RETURN
1340 FOR D = 1 TO 1000: NEXT D: RETURN

```

AWARI

Another game with a very rich history, AWARI is one of a series of pebble-in-pits games generally known under the name of "Mancala." In *GAMES OF THE WORLD*, Frederic V. Grunfeld (Ballantine Books, 1975) reports that mancala games "were played for thousands of years in Egypt, where boards have been found carved into the stone of the pyramid of Cheops and the temples at Luxor and Karnak. The game spread to Asia and Africa, where Arabs developed certain variations. African slaves brought mancala games to Surinam and the West Indies, where they survive unchanged. In some rural areas of Africa today, children play these age-old games on "boards" scooped out of the ground."



Grunfeld's book continues:

Traditionally, Wari is played for fun and prestige, not for money. In some regions, it also has a religious significance. In Surinam, mourners will sometimes play AWARI, a form of Wari, at a funeral, on the day before the corpse is to be interred. This is done to amuse the spirit which has not yet departed, but at sundown the boards are put away. It is believed that if the game is played at night, ghosts will join the living players and fly off with their spirits.

This computer program rendition of AWARI begins with six "pits" (the letters A to F and L to G) facing each player. Your pits are those from L to G. Each pit contains three seeds at the beginning of the game.

Choosing any pit on your side, you pick up all the seeds from it, and then proceed to move in a clockwise direction, sowing a seed in each pit as you go past it. You do not sow any seeds in the pits at either end of the board, the ones that start off as zeros.

If your final seed lands in an empty pit, then all the seeds in the opposite pit become yours, and are transferred to your "home." Your home is the zero to the left of the board, the computer's home is the zero to the right.


The game continues until either side is completely empty, so the player cannot move. At this point, the player with the largest number of seeds in their "home" is the winner. The computer plays well in this game, but with practice you'll learn to defeat it.



Reading AWARI

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN AWARI. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



AWARI

by Tim Hartnell
Giant Book of Computer Games
Ballantine Books

Copyright 1983, 1984. All rights reserved.

AWARI

```

10 REM AWARI
11 REM
12 REM FROM TIM HARTNELL'S
13 REM GIANT BOOK OF
14 REM COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 GOSUB 930: REM INITIALIZE
30 GOSUB 770: REM PRINT BOARD
40 FOR D = 1 TO 1000: NEXT D
50 GOSUB 160: REM MACHINE MOVE
60 GOSUB 770: REM PRINT BOARD
70 GOSUB 620: REM HUMAN MOVE
80 CW = 0: HW = 0
90 FOR C = 1 TO 12
100 IF C < 7 THEN CW = CW + A(C)

110 IF C > 6 THEN HW = HW + A(C)

120 NEXT C
130 IF CW = 0 OR HW = 0 THEN
510
140 GOTO 30
150 REM *****
160 REM MACHINE MOVE
170 GM = 0: C = 0: GW = 0
180 C = C + 1: IF C > 6 THEN 250
190 IF A(C) = 0 THEN 180
200 Z = C + A(C)
210 IF Z > 12 THEN Z = Z - 12
220 IF A(Z) = 0 AND A(13 - Z) >
    GW AND (13 - Z) < C THEN
    GM = C: GW = A(13 - Z)
240 IF C < 6 THEN 180
250 IF GM = 0 THEN 370
260 C = GM
270 PRINT "I'LL MOVE FROM " CHR$(
    64 + C) ".": FOR D = 1 TO 15
    00: NEXT D
280 FOR Z = C TO C + A(C)
290 IF Z > 12 THEN A(Z - 12) = A
    (Z - 12) + 1
300 IF Z < 13 THEN A(Z) = A(Z) +
    1
310 NEXT Z
320 Z = C + A(C) - 1: IF Z > 12 THEN
    Z = Z - 12
330 A(C) = 0
340 B(2) = B(2) + A(13 - Z): A(13 -
    Z) = 0
350 RETURN
360 REM *****
370 REM NON-SCORE MOVE
380 W = 0
390 W = W + 1
400 C = INT ( RND (1) * 6) + 1
410 IF A(C) < 0 THEN 440
420 IF W < 20 THEN 390
430 GOTO 510
440 PRINT "I'LL MOVE FROM " CHR$(
    64 + C) ".": FOR D = 1 TO 1
    500: NEXT D
450 FOR Z = C TO C + A(C)
460 IF Z < 13 THEN A(Z) = A(Z) +
    1
470 IF Z > 12 THEN A(Z - 12) = A
    (Z - 12) + 1
480 NEXT Z
490 A(C) = 0: GOTO 350
500 REM *****
510 REM END OF GAME
520 GOSUB 770
530 PRINT : PRINT "THAT'S THE EN
    D OF THE GAME."
540 PRINT
550 IF B(1) > B(2) THEN PRINT "
    YOU'RE THE WINNER!"
560 IF B(1) < B(2) THEN PRINT "
    AND I'M THE WINNER!"
570 IF B(1) = B(2) THEN PRINT "
    IT LOOKS LIKE A DRAW!"
580 PRINT : PRINT "MY SCORE WAS
    "B(2)
590 PRINT "AND YOURS WAS "B(1)
600 END
610 REM *****
620 REM HUMAN MOVE
630 PRINT "WHICH PIT TO START WI
    TH (G-L) " ;: GET A$: PRINT A
    $
640 B = ASC (A$) - 64
650 IF B < 7 OR B > 12 THEN 630
660 CO = B: Z = B + A(B): IF Z > 1
    2 THEN Z = Z - 12
670 M = A(Z)
680 FOR Z = B TO B + A(B)
690 IF Z > 12 THEN A(Z - 12) = A
    (Z - 12) + 1
700 IF Z < 13 THEN A(Z) = A(Z) +
    1
710 NEXT Z
720 Z = B + A(B) - 1: IF Z > 12 THEN
    Z = Z - 12
725 A(CO) = 0
730 IF M = 0 THEN B(1) = B(1) +
    A(13 - Z): A(13 - Z) = 0
750 RETURN
760 REM *****
770 REM PRINT BOARD
780 HOME : PRINT : PRINT : PRINT
790 HTAB 9: PRINT " A B C
    D E F": HTAB 13
800 FOR C = 1 TO 6
810 PRINT A(C); " ";
820 NEXT C
830 PRINT : HTAB 10: PRINT B(1)"
    "B(2): HTAB
    13
840 FOR C = 12 TO 7 STEP - 1
850 PRINT A(C); " ";
860 NEXT C
870 PRINT : HTAB 9: PRINT " L
    K J I H G"
880 PRINT : PRINT
890 RETURN
920 REM *****
930 REM INITIALIZE
940 HOME : VTAB 3: HTAB 17: PRINT
    "AWARI": VTAB 20: HTAB 9: PRINT
    "PRESS ANY KEY TO START " ;: GET
    A$
960 DIM A(12), B(2)
970 FOR C = 1 TO 12
980 A(C) = 3
990 NEXT C
1000 RETURN

```


GOMOKU

Considered to be one of the oldest games in existence, as well as the greatest strategic skill game, Go originated in China about 4,000 years ago. Go's background is summarized in *GAMES OF THE WORLD*, edited by Frederick V. Grunfeld (Ballantine Books, 1975):

For centuries after its introduction to Japan in 735 A.D., only the nobility played it. In due course the game spread to all levels of society and gifted players could attend Go academies. Masters merited special positions at the courts of Japanese feudal barons, or wandered the country giving lessons in the innumerable variations of play. Until 1600, Go was a compulsory course at the military academy of Japan.

Over the years, the basic concept of Go has been modified and refined with one off-shoot being GOMOKU. In translation, the name corresponds to GO (meaning five) and MOKU (meaning stones).



You'll find that GOMOKU is a relatively easy game to learn, but one that is almost impossible to totally master. The object is to get five of your pieces (H's) in a row, in any direction, while the computer is trying to do the same.

The playing board is an 8-by-8 grid. Enter your moves as a two-digit number with the vertical row number (1-8) first and then the horizontal number row (1-8). Thus, "81" is the lower left-hand corner and "88" is the lower right-hand.




The computer plays extremely well in this program, that is based on one written by Graham Charlton. It may take a couple of tries to fully understand all the subtleties in strategy, but once mastered, GOMOKU should keep you coming back for more.



Reading GOMOKU

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN GOMOKU. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



GOMOKU

by Tim Hartnell

Giant Book of Computer Games

Ballantine Books

Copyright 1983, 1984. All rights reserved.

—

GOMOKU

```

10 REM GOMOKU
11 REM
12 REM FROM TIM HARTNELL'S
13 REM GIANT BOOK OF
14 REM COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 GOSUB 750
30 GOSUB 130
40 GOSUB 240
50 GOSUB 130
60 GOSUB 320
70 GOSUB 130
80 IF L > 3 THEN PRINT : PRINT
   "I WIN!!": END
90 GOTO 40
100 E = A
110 E = E + N: IF A(E) < > Z THEN
   RETURN
120 K = K + 1: GOTO 110
130 HOME
140 PRINT : PRINT : PRINT
150 PRINT TAB( 12);"1 2 3 4 5 6
   7 8"
160 FOR A = 1 TO 8: PRINT TAB(
   10);A;" ";
170 FOR B = 2 TO 9
180 PRINT CHR$( A(A * 10 + B));
   ";
190 NEXT B
200 PRINT A
210 NEXT A
220 PRINT TAB( 12);"1 2 3 4 5 6
   7 8"
230 RETURN
240 PRINT : PRINT
250 PRINT "PLEASE ENTER YOUR MOV
   E: ";
260 INPUT G
270 G = G + 1
280 IF G < 12 OR G > 89 OR A(G) <
   > 46 THEN 260
290 Z = H
300 A(G) = Z
310 RETURN
320 A = G
330 L = 0
340 FOR X = 1 TO 4:K = 0:N = X(X
   )
350 GOSUB 100
360 N = - N: GOSUB 100
370 IF K > L THEN L = K
380 NEXT X
390 IF L > 3 THEN PRINT : PRINT
   "YOU WIN!!": END
400 T = 1
405 VTAB 18: HTAB 13: FLASH : PRINT
   " I'M THINKING ": NORMAL
410 IF T < > 2 THEN Z = C
420 IF T = 2 THEN Z = H
430 G = 0:H1 = 0:L = 0
440 FOR A = 12 TO 89
450 M = 0
460 IF A(A) < > 46 THEN 570
470 FOR X = 1 TO 4:K = 0:N = X(X
   )
480 GOSUB 100
490 N = - N: GOSUB 100
500 IF K > L THEN H1 = 0:L = K
510 IF L < > K THEN 540
520 IF T = 1 AND L < 4 OR (T = 2
   OR T = 3) AND L < 2 THEN 54
   0
530 M = M + 1
540 NEXT X
550 IF M < = H1 THEN 570
560 H1 = M:G = A
570 NEXT
580 IF H1 < > 0 THEN 650
590 T = T + 1: IF T < > 4 THEN 4
   10
600 A = 1
610 G = INT ( RND (1) * 77) + 13

620 IF A(G) = 46 THEN 650
630 A = A + 1: IF A < 100 THEN 61
   0
640 PRINT : PRINT "I CONCEDE THE
   GAME": PRINT "TO A MASTER!!
   ": END
650 A(G) = C
660 Z = C:A = G:L = 0
670 FOR X = 1 TO 4
680 K = 0
690 N = X(X)
700 GOSUB 100
710 N = - N: GOSUB 100
720 IF K > L THEN L = K
730 NEXT X
740 RETURN
750 HOME
760 DIM A(100),X(4)
770 FOR C = 1 TO 8
780 FOR B = 2 TO 9
790 A(C * 10 + B) = 46
800 NEXT B
810 NEXT C
820 FOR Q = 1 TO 4
830 READ Z:X(Q) = Z
840 NEXT Q
850 DATA 1,9,10,11
860 H = ASC ("H"):C = ASC ("C")

870 PRINT : PRINT "ENTER Y IF YO
   U WANT THE"
880 PRINT "FIRST MOVE, N IF YOU
   DON'T. ";
890 N = 0
900 N = N + 1
910 GET A$
920 IF A$ < > "Y" AND A$ < > "
   N" THEN 900
940 HOME
950 IF A$ = "Y" THEN RETURN
960 FOR J = 1 TO INT ( RND (1) *
   12)
970 READ Z
980 NEXT J
990 A(Z) = C
1000 RETURN
1010 DATA 34,35,44,46,47,54,55,
   56,57,66,65,67

```

LAS VEGAS HIGH


Time to take your chances with our one-armed bandit, straight from the old West. LAS VEGAS HIGH is a slot-machine simulation. You need only decide how much you'll bet and the reels will whirl away.



Your winnings — as you'd expect — are based on your bet and the relative difficulty of the various combinations coming up. Here are some sample odds:

Three bars:	9 times your bet, Jackpot!
Three bells:	3.9 times your bet
Three others:	3.5 times your bet
Two of a kind:	1.7 times your bet


You also win a bonus for getting:



Bell at the end
Apple at each side
Cherry, Bell, Cherry

The computer keeps up the chatter as the game, and your wealth (or poverty) unfold. Win enough and you can break the bank.


You start with a bankroll of \$250, and you can bet all or part of it. Press the SPACEBAR to run the game. Your winnings are automatically added to your bank. Exit early by pressing CONTROL-RESET.



Reading LAS VEGAS HIGH

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter `RUN LAS.VEGAS.HIGH`. Operating instructions are found in the article. Exit anytime by pressing `CONTROL-RESET`.



LAS VEGAS HIGH

by Tim Hartnell

Giant Book of Computer Games

Ballantine Books

Copyright 1983, 1984. All rights reserved.

LAS VEGAS HIGH

```

10 REM LAS VEGAS HIGH
11 REM
12 REM FROM TIM HARTNELL'S
13 REM GIANT BOOK OF
14 REM COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 GOSUB 1110: REM INITIALIZE
30 GOSUB 870: REM PLAYER INPUT
40 GOSUB 520: REM OPERATE SLOT M
  ACHINE
50 IF CASH < 1 THEN 90
60 IF CASH > 2500 THEN 290
70 GOTO 30
80 REM *****
90 REM BROKE
100 REM *****
110 GOSUB 410
120 PRINT "THAT'S THE END OF THE
  LINE,"
130 PRINT "OH ONCE MIGHTY GAMBLE
  R..."
140 GOSUB 410
150 PRINT "YOU'RE STONE, FLAT BR
  OKE!!"
160 GOSUB 410
170 PRINT "PRESS 'Y' IF YOU'D LI
  KE TO"
180 PRINT "HAVE ANOTHER GO AT BR
  EAKING"
190 PRINT "!!!!!! LAS VEGAS HIGH
  !!!!!"
200 PRINT "(OR PRESS 'N' IF YOU
  WISH TO LEAVE) ";
210 GET A$
220 IF A$ < > "Y" AND A$ < > "
  N" THEN 210
230 IF A$ = "Y" THEN RUN
240 PRINT : PRINT "OK, PUNTER..."
  "
250 GOSUB 410
260 PRINT "THANKS FOR THE GAME!"

270 END
280 REM *****
290 REM BROKE THE BANK
300 REM *****
310 GOSUB 410
320 PRINT "WELL DONE, GAMBLER!!"

330 GOSUB 410
340 PRINT "YOU'VE REACHED OUR HO
  USE LIMIT"
350 PRINT "SO WE'LL HAVE TO THRO
  W YOU OUT"
360 GOSUB 410
370 PRINT "PEOPLE WITH LUCK LIKE
  YOURS GIVE"
380 PRINT "OUR CASINO A BAD NAME
  ......."
390 GOTO 160
400 REM *****
410 REM DELAY
420 REM *****
430 FOR P = 1 TO 1000: NEXT P
440 PRINT : PRINT
450 RETURN
460 REM *****
470 REM DELAY TWO
480 REM *****
490 FOR P = 1 TO 1000: NEXT P
500 RETURN
510 REM *****
520 REM OPERATE SLOT MACHINE
530 REM *****
540 HOME
550 GOSUB 410
560 PRINT "/^~^~^~^~^~^~^~^~^~^
  ~^~^~^~^~^~^~^~^~^\"
570 PRINT " * ";
580 FOR M = 1 TO 3
590 GOSUB 460
600 A = INT ( RND (1) * 43)
610 IF A < 2 THEN PRINT A$(4);:
  C(M) = 1
620 IF A > 1 AND A < 6 THEN PRINT
  A$(3);:C(M) = 2
630 IF A > 5 AND A < 12 THEN PRINT
  A$(1);:C(M) = 3
640 IF A > 11 AND A < 20 THEN PRINT
  A$(2);:C(M) = 4
650 IF A > 19 AND A < 31 THEN PRINT
  A$(5);:C(M) = 5
660 IF A > 30 THEN PRINT A$(6);
  :C(M) = 6
670 PRINT " * ";
680 NEXT M
690 GOSUB 410
700 WIN = 0

```

```

710 IF C(1) + C(2) + C(3) = 3 THEN
    PRINT "THREE BARS!!!": GOSUB
410: PRINT "THAT'S JACKPOT S
TYLE!!":WIN = WIN + 9: GOTO
750
720 IF C(1) = C(2) AND C(3) = C(
2) AND C(1) = 2 THEN PRINT
"THREE BELLS!!!":WIN = WIN +
3.9: GOTO 750
730 IF C(1) = C(2) AND C(3) = C(
2) AND C(1) < > 1 AND C(2) <
> 3 THEN PRINT "THREE OF A
KIND":WIN = WIN + 3.5: GOTO
750
740 IF C(1) = C(2) OR C(1) = C(3
) OR C(2) = C(3) THEN PRINT
">> A PAIR <<":WIN = WIN + .
7
750 IF C(3) = 2 THEN PRINT "A B
ELL AT THE END IS A BONUS!":
WIN = WIN + .6
760 IF C(1) = 3 AND C(3) = 3 THEN
    PRINT "AN APPLE AT EACH SID
E IS GOOD":WIN = WIN + .5
770 IF C(1) = 4 AND C(2) = 3 AND
C(3) = 4 THEN PRINT "THAT O
LD 'CHERRY, BELL, CHERRY': PRINT
"COMBINATION IS ONE OF MY FA
VORITES!":WIN = WIN + .4
780 GOSUB 410
790 WIN = INT (BET * WIN)
800 IF WIN > 0 THEN PRINT "AND
YOU'VE WON $"WIN!":CASH = C
ASH + WIN
810 IF WIN = 0 THEN PRINT "AND
YOU'VE LOST $"BET:CASH = CAS
H - BET
820 GOSUB 410
830 IF CASH > 0 THEN PRINT "SO
YOU NOW HAVE $"CASH
840 GOSUB 460
850 RETURN
860 REM *****
870 REM PLAYER INPUT
880 REM *****
890 HOME
900 GOSUB 410
910 IF CASH < 300 THEN PRINT "H
I THERE, GAMBLER!"
920 IF CASH > 299 AND CASH < 600
THEN PRINT "YOU'RE DOING W
ELL TONIGHT!"
930 IF CASH > 599 AND CASH < 900
THEN PRINT "LADY LUCK HAS
CERTAINLY SMILED ON YOU!"
940 IF CASH > 899 AND CASH < 120
0 THEN PRINT "THE FATES ARE
BEING EXTREMELY KIND"
950 IF CASH > 1199 THEN PRINT "
IT IS SO GOOD TO SEE AN EXPE
RT AT WORK"
960 GOSUB 470
970 PRINT : PRINT "YOU HAVE $"CA
SH
980 PRINT : INPUT "HOW MUCH DO Y
OU WANT TO BET? ";BET
990 IF BET > CASH THEN PRINT "Y
OU AIN'T GOT THAT MUCH!": GOTO
980
1000 GOSUB 410
1010 PRINT "OK, SIR, $"BET" IT I
S!"
1020 GOSUB 410
1030 PRINT "PRESS THE SPACEBAR T
O PLAY ";
1040 GET A$: PRINT : PRINT
1050 FOR T = 1 TO 40
1060 PRINT SPC( T / 2);"***** S
TAND BY *****"
1070 PRINT
1080 NEXT T
1090 RETURN
1100 REM *****
1110 REM INITIALIZE
1120 REM *****
1130 HOME
1140 DIM A$(6),C(6)
1160 CASH = 250
1170 FOR B = 1 TO 6
1180 READ A$(B)
1190 NEXT B
1200 RETURN
1210 DATA "$ APPLE $","# CHERRY
#"," * BELL *","!! BAR !!","
< LEMON >"," { PLUM }"

```

REVERSI/OTHELLO™

REVERSI was invented in the late eighteen-hundreds and is played on an ordinary eight-by-eight playing board. You use pieces that have different colors on each side. The game begins with four pieces placed on the center squares.

From this point on, you move by placing one of your pieces next to a computer piece or pieces, with another of your pieces further on. When that happens, all the computer pieces “reverse” to become your pieces.



Here's how it works. Suppose a line of pieces looked like this:

OXXXX

and you decided to put your piece (the O) at the end of the line like this:

OXXXXO



The computer pieces would reverse, so the line would look like this after your move:

OOOOOO

The game continues until every square on the board is filled, or neither player can move. As you can see, fortunes can change swiftly in this game, as rows branching off your position (such as the diagonals) can be changed with a single move.

Enter your move as a two-digit number, where the first number is squares down from the top and the second number is squares to the right. Thus, 81 is the lower left corner and 18 is in the upper right. If you cannot move at any time, enter a zero. Exit early by pressing CONTROL-RESET.

REVERSI is an enjoyable game to play. Writing in *Creative Computing* magazine (June, 1981, p. 188), David Levy comments that it is "one of the best games ever invented, simply because the rules can be learned in no more than a minute, yet the game can take years to master."

Back in the 1880's, when REVERSI was first invented in London, two gentlemen — Lewis Waterman and John W. Mollett — both claimed to be the originators of the game. Stephen Kimmel (writing in *Creative Computing*, July, 1981, p. 94) believes that Waterman has the stronger case for being the inventor, because he had published the details first in a series of articles in the magazine, *The Queen*.

A modern version of the game is OTHELLO (a registered trademark of Gabriel, Industries, Inc.), developed by Goro Hasegawa in Tokyo in 1971. The only difference between it and REVERSI is that the first four positions are determined by the rules of the game.


Regardless of who should get the credit, Waterman, Mollett, or Hasegawa, it is a great game and this program puts up a spirited defense.



Reading REVERSI

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN REVERSI. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



REVERSI

by Tim Hartnell

Giant Book of Computer Games

Ballantine Books

Copyright 1983, 1984. All rights reserved.

REVERSI/OTHELLO

```

10 REM REVERSI
11 REM
12 REM FROM TIM HARTNELL'S
13 REM GIANT BOOK OF
14 REM COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 GOTO 740
30 PRINT "MY MOVE..."
40 S = O:T = X:H = 0
50 FOR A = 2 TO 9: FOR B = 2 TO
  9
60 IF A(A,B) < > 46 THEN 210
70 Q = 0
80 FOR C = - 1 TO 1: FOR D = -
  1 TO 1
90 K = 0:F = A:G = B
100 IF A(F + C,G + D) < > S THEN
  130
110 K = K + 1:F = F + C:G = G + D

120 GOTO 100
130 IF A(F + C,G + D) < > T THEN
  150
140 Q = Q + K
150 NEXT D: NEXT C
160 IF A = 2 OR A = 9 OR B = 2 OR
  B = 9 THEN Q = Q * 2
170 IF A = 3 OR A = 8 OR B = 3 OR
  B = 8 THEN Q = Q / 2
180 IF (A = 2 OR A = 9) AND (B =
  3 OR B = 8) OR (A = 3 OR A =
  8) AND (B = 2 OR B = 9) THEN
  Q = Q / 2
190 IF Q < H OR ( RND (1) < .3 AND
  Q = H) THEN 210
200 H = Q:M = A:N = B
210 NEXT B: NEXT A
220 IF H = 0 AND R = 0 THEN 690
230 IF H = 0 THEN 250
240 GOSUB 580
250 GOSUB 370
260 INPUT "ENTER YOUR MOVE (0 TO
  PASS) ";R
270 REM ENTER 0 TO PASS
280 S = X:T = O: REM LETTER O
290 IF R = 0 THEN 350
300 IF R < 11 OR R > 88 THEN 260

310 R = R + 11
320 M = INT (R / 10)
330 N = R - 10 * M
340 GOSUB 580
350 GOSUB 370
360 GOTO 30
370 REM PRINT BOARD
380 C = 0:H = 0
390 HOME
400 PRINT
410 HTAB 13: PRINT "COMPUTER IS
  X"
420 HTAB 15: PRINT "HUMAN IS O"
430 PRINT
440 PRINT TAB( 12);"1 2 3 4 5 6
  7 8"
450 FOR B = 2 TO 9: PRINT TAB(
  10);B - 1;" ";
460 FOR D = 2 TO 9
470 PRINT CHR$( A(B,D));" ";
480 IF A(B,D) = X THEN C = C + 1

490 IF A(B,D) = O THEN H = H + 1

500 NEXT D
510 PRINT B - 1
520 NEXT B
530 PRINT TAB( 12);"1 2 3 4 5 6
  7 8"
540 PRINT : PRINT
550 HTAB 14: PRINT "COMPUTER: "C

560 PRINT : HTAB 14: PRINT "  H
  UMAN: "H
570 PRINT : RETURN
580 FOR C = - 1 TO 1: FOR D = -
  1 TO 1
590 F = M:G = N
600 IF A(F + C,G + D) < > S THEN
  630
610 F = F + C:G = G + D
620 GOTO 600
630 IF A(F + C,G + D) < > T THEN
  670
640 A(F,G) = T
650 IF M = F AND N = G THEN 670
660 F = F - C:G = G - D: GOTO 640

670 NEXT D: NEXT C
680 RETURN
690 GOSUB 370
700 IF C > H THEN PRINT "I'M TH
  E CHAMP!"
710 IF H > C THEN PRINT "YOU'RE
  THE CHAMP!"
720 IF H = C THEN PRINT "IT'S A
  DRAW!"
730 END
740 HOME
745 VTAB 5: HTAB 16: PRINT "REVE
  RSI"
750 X = ASC ("X"):O = ASC ("O")
  : REM LETTER O NOT ZERO
760 DIM A(10,10)
770 FOR B = 1 TO 10: FOR C = 1 TO
  10
780 IF B < > 1 AND C < > 1 AND
  B < > 10 AND C < > 10 THEN
  A(B,C) = ASC (".")
790 NEXT C: NEXT B
800 A(5,5) = X:A(6,6) = X
810 A(6,5) = O:A(5,6) = O
820 P = 0
830 PRINT : PRINT : PRINT "DO YO
  U WANT THE FIRST MOVE"
840 PRINT ,(Y OR N) ";
850 GET W$
860 GOSUB 370
870 IF W$ = "Y" THEN GOTO 260
880 GOTO 30

```

CAUZIN'S CORNER...

and now for something slightly different

These five games are written in BASIC, so you can see how they work and modify them. To see the commands, LOAD the program and enter LIST. You'll see all the program lines scroll down the screen. Enter LIST 200 to see just one line, in this case line 200.

If you want a printout of the program, enter PR#1 and then enter LIST. It will be useful to have a printout to help you learn more about BASIC and about these games.

Some moves are determined by random chance. In BASIC this is the RND(1) function. To improve your chances of winning or to make the game harder, look for the use of this command (especially in LAS VEGAS HIGH and AWARI) and change the odds.

The game AWARI uses line 400 to determine a random move for the computer.

$$400 \text{ C} = \text{INT}(\text{RND}(1) * 6) + 1$$

The RND statement multiplied by six gives a random number from .0001 to 5.9999. The INT function rounds it off to 0-5. When you add 1, the range becomes 1-6. If the program cannot find a winning move, it uses this line to randomly pick one of the six pits. To handicap the computer, change the 6 to 5 or 4 and limit the computer's choices.

These games all use the text screen. You can use the programs as algorithms, or models, and add your own graphics and sounds. This can be as simple or complicated as you care to make it.

For example, try using "text graphics" to add color to GOMOKU (the same techniques can be used for REVERSI). These are graphics created with a PRINT statement while in lo-res (GR) mode. After you LOAD the program, type in the following lines. They will automatically take the place of the previous ones. Save the modified program under a new name, to preserve the original version.

```
140 REM
150 REM
160 FOR A = 1 TO 8
175 VTAB A * 2:HTAB (B * 2) + 10
180 PRINTCHR$(A(A * 10 + B))
200 PRINT
240 VTAB 21
280 IF G < 12 OR G > 89 OR A(G) <> 76 THEN 260
320 REM
390 IF L > 3 THEN TEXT:HOME:PRINT
    "YOU WIN!":END
405 VTAB 21:HTAB 15:FLASH:PRINT
    " I'M THINKING ":NORMAL
460 IF A(A) <> 76 THEN 570
620 IF A(G) = 76 THEN 650
635 TEXT:HOME
790 A(C * 10 + B) = 46
860 H = 127:C = 59
930 GR
```


Line 930 puts your Apple into lo-res mode and clears the screen. The characters of the board are PRINTed by lines 160-200. Line 790 and 860 define the characters that produce effective colors. Try using other numbers for different color combinations. Text graphics might be a way for you to add easy color graphics and animation to many of your own programs.

Introducing **StripWare**TM “The Best of Both Worlds”

There's a world of Softstrip data strips coming your way. Besides being in magazines and books, data strips are now available in many exciting Cauzin StripWare titles.

StripWare offers a wide range of the best programs from some of the world's leading computer magazines, books, and authors.

FOR YOUR APPLE

Art and Graphics on the
Apple II/IIe

BASIC Apple IIc

Home Education

Computer Puzzles

Essential Programs for
Small Business Planning

**Titles are also
available for
IBM PC and
Macintosh**




Cauzin Systems, Inc.
835 Main Street
Waterbury, CT 06706

From board games to brain games, instructions for programming 50 exciting games in BASIC. Includes:

- Board Games—Checkers & Othello
- Simulations—like Mistress of Xenophobia
- Dice Games—Snakes, One and Twenty
- Unique Games—like Proboscidean, computerized "Concentration"


Features tips on how to create your own games, how to modify standard game programs, and a glossary of common computer terms.



An all new collection of more than 40 original games to program in BASIC. Includes:

- Board Games
- Brain Teasers
- Space Games
- Simulations


With tips on how to create your own games and modify existing games, this book is all you need to have fun (with a computer).



The master gamesman shows how to program four basic games—from the simple "Werewolves and Wanderers" to more sophisticated adventures—and teaches you adventure addicts everything you need to know to:

- Construct adventure worlds and maps
- Create labyrinths, magic spells & daring heroes
- Keep track of player's movements
- Design battles with hideous monsters

All programs are written in the BASIC computer language.



Ballantine Books, Dept. SH, 201 E. 50th St., New York, N.Y. 10022

Please send me _____ copies **TIM HARTNELL'S GIANT BOOK OF COMPUTER GAMES @ \$8.95**
 _____ copies **TIM HARTNELL'S SECOND GIANT BOOK OF COMPUTER GAMES @ \$9.95**
 _____ copies **CREATING ADVENTURE GAMES ON YOUR COMPUTER @ \$9.95**

Enclosed is a check or money order for \$_____, including \$1.00 per book for postage and handling, and tax where applicable. (Sorry, no cash or C.O.D.s.)

Name _____

Address _____

City _____ State _____ Zip _____

Allow at least 4 weeks for delivery. Prices are subject to change without notice.

LIMITED WARRANTY

CAUZIN SYSTEMS, INC. (CAUZIN) warrants this package against defects in materials and workmanship for a period of 90 days from the date of original purchase.

If you discover a defect, CAUZIN will, at its option, either repair, replace, or refund the purchase price of this product at no charge to you, provided it is returned during the warranty period indicated above or any extended period(s) paid for by the user. Transportation charges should be prepaid, to the authorized dealer from whom you purchased it or to any other authorized CAUZIN dealer in the country.

You may obtain additional information from CAUZIN directly at the address printed on this certificate, or by calling (203) 573-0150. Please attach your name, address, telephone number, a description of the problem, and a copy of a bill of sale as proof of date of original purchase, to each product returned for warranty service.

This warranty applies only to products manufactured by or for CAUZIN, which can be identified by the CAUZIN

Cauzin StripWare™

trademark, trade name, or logo affixed to the product. CAUZIN does not warrant any product(s) that are not CAUZIN products. This warranty does not apply if the product has been damaged by accident, abuse, misuse or misapplication, has been modified without the written permission of CAUZIN, or if any CAUZIN serial number has been removed, altered, or defaced.

ALL WARRANTIES, EXPRESSED OR IMPLIED, INCLUDING WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE, ARE LIMITED IN DURATION TO 90 DAYS FROM THE ORIGINAL DATE OF PURCHASE. THERE ARE NO WARRANTIES WHICH EXTEND BEYOND THE DESCRIPTION ON THE FACE HEREOF.

THE WARRANTY AND REMEDIES SET FORTH ABOVE ARE EXCLUSIVE AND IN LIEU OF ALL OTHERS, ORAL OR WRITTEN, EXPRESSED OR IMPLIED. NO CAUZIN DEALER, AGENT, OR EMPLOYEE IS AUTHORIZED TO MAKE ANY MODIFICATION, OR ADDITION TO THIS


835 South Main Street
Waterbury, Connecticut 06706

WARRANTY.

CAUZIN IS NOT RESPONSIBLE FOR SPECIAL, INCIDENTAL, OR CONSEQUENTIAL DAMAGES RESULTING FROM ANY BREACH OF WARRANTY, OR UNDER ANY OTHER LEGAL THEORY, INCLUDING LOST PROFITS, DOWNTIME, GOODWILL, DAMAGE TO, OR REPLACEMENT OF EQUIPMENT AND PROPERTY, AND ANY COST OF RECOVERING, REPROGRAMMING OR REPRODUCING ANY PROGRAM OR DATA STORED IN OR USED WITH CAUZIN PRODUCTS.

Somes states do not allow the exclusion or limitation of incidental or consequential damages or limitations on how long an implied warranty lasts, so the above limitations or exclusion may not apply to you.

This warranty shall not be applicable to the extent that any provision of this warranty is prohibited by an federal, state or municipal law which cannot be preempted. This warranty gives you specific legal rights, and you may also have other rights which vary from state to state.



GET ORGANIZED!

Order your Cauzin Softstrip™ System high quality binder today! Designed especially to hold Cauzin StripWare™ packages, Communication program disk, and your data disk.

This attractive binder will help you keep track of your StripWare, disks, and makes a lovely addition to your bookshelf.

Available only through:
Cauzin Systems, Inc.
835 Main Street
Waterbury, CT 06706

Send check or money order for \$12 to cover cost, postage and handling.

- Always someone to challenge in
CHECKERS
- Discover the ancient strategy of
AWARI
- Five-in-a-row wins the game of
GOMOKU
- Try the **LAS VEGAS HIGH**
slot machine
- The game constantly shifts in
REVERSI/OTHELLO™

CAUZIN SYSTEMS, INC.
835 South Main Street
Waterbury, Conn. 06706
(203) 573-0150